

# Inequality: the Gini Index

## Goal

This exercise introduces the Gini index, which is one of the most commonly used income inequality indicators.

## Activity

Calculate the Gini index on total disposable income for Finland and the US in 2000, after bottom-coding disposable income at 1 percent of its equivalised mean and top-coding at 10 times the unequivalised median.

## Guidelines

- Prepare the data as you did in the previous exercise (drop observations with missing or zero *dpi*).
- In the previous exercise you have seen two different methods of dealing with extreme values, trimming and winsorising (or bottom-/top-coding). The LIS key figures are calculated using a particular type of bottom-/top-coding, which we will replicate in this exercise. The bottom-coding is carried out after the equivalisation of income (on the equivalised income distribution), while the top-coding is carried out before (on the unequivalised distribution) in the following way:

- at the bottom of the distribution, equivalised income is bottom-coded at 1 percent of its equivalised mean, i.e., all observations for which equivalised income is lower than 1% of the average equivalised income are set to that value.

```
qui sum ey [w=hweight*d4]
replace ey=0.01*r(mean) if ey<0.01*r(mean)
```

- at the top of the distribution, income is top-coded at 10 times the unequivalised median, i.e., all observations for which unequivalised income (or dpi) is higher than 10 times the median unequivalised income are set to that value.

```
qui sum dpi [w=hweight*d4], de
replace ey=(10*r(p50)/(d4^0.5)) if dpi>10*r(p50)
```

- Stata provides ado files that will calculate the Gini coefficient as well as several other inequality indices. One such command is:

```
ineqdeco [varname] [w=<weight>]
```

**Program**

```
di "*** INCOME DISTRIBUTION II - Exercise 13 ***"

program define bottop
  qui sum ey [w=hweight*d4]
  replace ey = .01*r(mean) if ey<.01*r(mean)
  qui sum dpi [w=hweight*d4], de
  replace ey = (10*r(p50)/(d4^.5)) if dpi>10*r(p50)
end

foreach file in $us00h $fi00h {
  display "`file'"
  use hweight d4 dpi if (!mi(dpi) & !(dpi==0)) using "`file'", clear
  gen ey=dpi/(d4^0.5)
  bottop
  ineqdeco ey [w=hweight*d4]
}
```

**Results**

	<b>Gini</b>
<b>US 2000</b>	<i>0.36823</i>
<b>Finland 2000</b>	<i>0.24621</i>

**Comments**

- The Gini index ranges between 0 and 1, with inequality increasing with an increasing index. A value of 0 means there is a completely equal distribution of income; a 1 refers to the extreme situation of one household holding the total population income, and all the rest having no income at all.
- As expected, inequality is much larger in the US than in Finland.
- To see the Ginis for all LIS datasets online, go to <http://www.lisproject.org/key-figures/key-figures.htm>.